

第3章

Maxima 入門 3

■既に習ったこと

- 四則演算 : `a+b;`, `a-b;`, `a*b;`, `a/b;`, `a^2;`
- 代入 : `ev(a^2, a:b+1);`
- 微分 : `diff(3*a^2,a);`
- 連立方程式 : `solve([2*a+3*b=21, 4*a-2*b=2],[a,b]);`
- 展開 : `expand((a+b+c)^2);`
- ある変数でまとめる : `factorout(a^2 +b^2 +c^2 +2*a*b +2*b*c +2*a*c, b, c);`
- 因数分解 : `factor(a^2 -b^2);`
- 簡単化 : `ratsimp(a^2+2*a*b+b^2+a^2+2*a*b+b^2);`
- 実数化 : `float(13*a/137);`
- 出力の省略 : `1+2$`
- 式の参照 : `expand((a+b+c)^2)$ factorout(%, b, c);`
- メモ : `/* クールノー競争 */`
- ギリシャ文字 : `%alpha;`
(Maxima のバージョンが新しい場合, `[Esc] a [Esc]` の順に入力しても良い。)
- 文字の定義 : `p:a-q;`
- 定義の解除 : `kill(p);, kill(all);`

3.1 対称な企業が n 社存在する場合

企業数が n である場合を分析できるようになると、企業数の変化（競争の変化）が市場に与える影響を分析できるようになる。ここでは、対称な企業が n 社存在する市場での同質財数量競争を分析してみよう。

企業 i ($i = 1, 2, \dots, n$) の生産量を x_i とする。企業 i 以外の企業の実生産量の和を $X_i = x_1 + x_2 + \dots + x_{i-1} + x_{i+1} + \dots + x_n = \sum_{k \neq i} x_k$ で表すことにする。市場全体の総生産量は $X = \sum_{k=1}^n x_k$ で表すことにする。各企業の限界費用は c で一定としよう。企業が

直面している逆需要関数は $p = 1 - X$ で与えられるとする。

このとき、企業 i の利潤は次式で与えられる。

$$\begin{aligned}\pi_i &= (p - c)x_i, \\ &= \left(1 - \sum_{k=1}^n x_k - c\right)x_i, \\ &= \left(1 - x_i - \sum_{k \neq i} x_k - c\right)x_i, \\ &= (1 - x_i - X_i - c)x_i.\end{aligned}$$

Maxima を用いてプログラミンする場合、この式の最後の行のような定義を用いると便利である。その理由は、企業 i の利潤最大化条件を考える場合、 π_i を x_i で偏微分することになるが、 x_i を含む部分とそうでない部分に分けておくためである。また、利潤最大化条件を求めた後で、企業の対称性を用いる場合にも、このような表記は有用となる。

ひとまず、この n 企業の数量競争の問題を解いてみよう。企業 i の利潤最大化条件より、次式が成立する。

$$\frac{\partial \pi_i}{\partial x_i} = 1 - 2x_i - X_i - c = 0, \quad (i = 1, 2, \dots, n).$$

ここで、企業の対称性より、均衡では $x^* = x_1 = \dots = x_n$ が成立しているはずであり、 $X_i = \sum_{k \neq i} x_k = -x_i + \sum_{k=1}^n x_k$ であることから、次式が成立する。

$$\frac{\partial \pi_i}{\partial x_i} = 1 - 2x_i - X_i - c = 0, \quad (3.1)$$

$$1 - 2x^* - (-x^* + nx^*) - c = 0, \quad (3.2)$$

$$1 - 2x^* - (n-1)x^* - c = 0, \quad (3.3)$$

$$1 - c = (n+1)x^*, \quad (3.4)$$

$$x^* = \frac{1-c}{n+1}. \quad (3.5)$$

この計算を Maxima でプログラムすると、以下ようになる。

```
X:xi+Xi$
p:1-X$
pii:(p-c)*xi$

diff(pii,xi)$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
solve( [%=0 ], [x] )$
ratsimp(%) ;
```

■均衡での値の図示 また、得られた価格や利潤を図示することもできる。横軸を n 、縦軸をこれらの値としたい場合、限界費用の値に何らかの数値を代入する必要がある。ここでは、 $c = 0$ とし、均衡価格や均衡利潤を図示してみよう。

図を描く準備として、先ほどの計算結果を持つ変数を定義しよう。ここでは CournotOutcome という変数名にしてみよう。これを行うには、先ほどのプログラムの最後の部分を書き換えれば良い。

```
diff(pii,xi)$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
solve([%=0],[x])$
ratsimp(%)$
CournotOutcome:%;
```

では、二次元平面上に均衡生産量を描いてみよう。二次元平面上にグラフを描く場合、関数「plot2d」を用いる。例えば、関数 x^2 を $0 \leq x \leq 5$ の範囲で図示したい場合は、以下の様なプログラムを入力すれば良い。

```
plot2d(x^2,[x,0,5]);
```

つまり、plot2d の中身の最初の部分に描きたい関数を入力し、その後ろの部分にその関数の定義域を入力すれば良いことになる。関数 $f(x)$ を定義域 $z1 < x < z2$ で描きたければ、plot2d($f(x)$, [x, z1 ,z2]); と入力すれば良い。

では、均衡生産量が企業数とともにどのように変化するかを図示してみよう。そのためには、次のようなプログラムを書けば良い。

```
x$
ev(%, CournotOutcome)$
ev(%, c:0)$
plot2d(%, [n,1,10]);
```

■自由参入市場 先ほどのモデルを発展させ、自由参入市場を分析してみよう。自由参入市場のモデルでは、まず企業たちが市場に参入するかを決定する（ステージ 1）。参入した場合、参入費用 f を支払わなければならないとする。参入企業数が決まった後で、参入した企業たちが数量競争を行う（ステージ 2）。ここでは、参入を決定した企業数を n 、企業 i の生産量を x_i 、企業の限界費用を c 、参入した企業が直面する逆需要関数を $p = 1 - \sum_{i=1}^n x_i$ と仮定して、このモデルを分析してみよう。

モデルに含まれるプレーヤー（経済主体）の行動の時点がずれている場合、後ろの時点から解くので、企業が参入した後の数量競争から計算してみよう。しかし、これは前の例とほぼ同じであり、参入企業の利潤に参入費用を含めただけである。したがって、この数

量競争の解を求めるプログラムは次のようになる。

```
kill(all)$

X:xi+Xi$
p:1-X$
pii:(p-c)*xi-f$

diff(pii,xi)$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
solve([%=0],[x])$
ratsimp(%)$
CournotOutcome:%;
```

次に、企業が参入するかを決める問題を考える。企業は参入後の利潤が正であるなら参入するはずなので、まず参入後の利潤を求める。また、企業数が増えると競争が激しくなり、利潤が減少するはずなので、参入企業数が増え続けると、いずれは参入企業の利潤がゼロになる。この企業の利潤をゼロにする企業数を実現すると、それ以上の参入は起こらず、また既に参入している企業も退出しないと考えられる。したがって、均衡企業数を求めたければ、企業の利潤のゼロと置き、それを企業数 n について解けば良いことになる。

これを行うためには、以下のプログラムを書けば良い。

```
pii$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
ev(%, CournotOutcome)$
solve([%=0],[n])$
ratsimp(%);
```

ここで、企業数は正であり、企業の限界費用 c は逆需要関数の切片 1 より小さい ($1 > c$) はずなので、均衡企業数は以下となる。

$$n^* = -\frac{\sqrt{f} + c - 1}{\sqrt{f}} (> 0).$$

得られた均衡企業数を `FreeEntry` と名付けるには、出力結果の左側の解を選択しなければならない。これはリストの選択と呼ばれる操作であるが、ここでは、とりあえず以下のプログラムを実施し、この操作を行うことにしよう*1。

*1 リストの操作については、別の資料で解説する予定である。

```

pii$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
ev(%, CournotOutcome)$
solve( [%=0], [n] ) [1]$
ratsimp(%)$
FreeEntry:n%;

```


■過剰参入定理 企業が自分の意志で参入し競争を行うことは、社会にとって良さそうな感じがする。この直観が正しいのかを確かめてみよう。そのためには、社会の良さを測る指標が必要となる。経済学では多くの場合、この指標として社会厚生 SW を用いる。社会厚生は消費者余剰 CS と生産者余剰 PS の和で定義される。消費者余剰は、需要曲線、均衡価格を通る水平な線、縦軸が囲む面積であるので、それは図 3.1 のようになる。また、今回のモデルでは生産者余剰は企業の利潤の和に一致するので、社会厚生 SW 、消費者余剰 CS 、生産者余剰 PS は次式で与えられる。

$$CS = \frac{(1 - p^*)X^*}{2} = \frac{(X^*)^2}{2}, \quad (3.6)$$

$$PS = n(p^* - c)x^* - nf = n\pi_i^*, \quad (3.7)$$

$$SW = CS + PS. \quad (3.8)$$

図 3.1 社会厚生，消費者余剰，生産者余剰



これらを Maxima で定義すると、次式のようになる。

```
X^2/2$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
ev(%, CournotOutcome)$
ratsimp(%)$
CS:%;
```

```
n*pii$
ev(%,xi:x)$
ev(%,Xi:(n-1)*x)$
ev(%, CournotOutcome)$
ratsimp(%)$
PS:%;
```

```
CS+PS$
SW:ratsimp(%) ;
```

ここで、先ほど求めて均衡企業数 n^* が社会厚生を最大化しているのであれば、社会厚生を微分したものに n^* を代入すれば、ゼロになるはずである。逆に、この値が正であれば、均衡企業数よりも企業数を増やすと社会厚生が増え、この値が負であれば、均衡企業数よりも企業数を減らすと社会厚生が増えることとなる。どの状態が実現しているのかを求めてみるには、次のプログラムを実行すれば良い。

```
SW$
diff(%,n)$
ev(%,FreeEntryn)$
factor(%) ;
```

すると、次のような結果を得る。

$$\frac{\partial SW(n^*)}{\partial n} = \frac{(\sqrt{f} + c - 1) f}{1 - c} (< 0).$$

ここで、分子は負となっていることに注意してほしい。これは、均衡企業数の値が正となるために必要な条件であった。以上から、企業が自由に参入できる使用において、均衡企業数からわずかに企業数を増やすと、社会厚生は減少することがしめされた。逆に言うと、均衡企業数より少ない企業数を実現させることができれば、社会厚生は増加することになる。このように、自由参入の下では、社会的に望ましい企業数よりも多い企業数を実現してしまうことを過剰参入定理と呼んでいる。